Statistical Bulletin
Reliability & Variation Research

DETROIT RESEARCH INSTITUTE
P.O. BOX 36504 • GROSSE POINTE, MICHIGAN 48236 • (313) 886-7976

LEONARD G. JOHNSON
EDITOR

WANG H. YEE
DIRECTOR

Volume 21 November , 1991
Bulletin 6 Page 1

USING LIFE TEST DATA TO EVALUATE A DESIGN'S COMPLIANCE TO A SPECIFIED WEIBULL GOAL LINE

INTRODUCTION

Every design for a part or sub-assembly in a system has what is called an acceptable level of usage reliability when in the hands of the customer who buys the system from a manufacturer. This acceptable or desirable level of reliability can be defined by a distribution function of life for the system.

In recent years the Weibull Distribution has come to be very frequently used as the defining criterion from the standpoint of specifying a product reliability goal as far as service life is concerned. Consequently, it is important that a manufacturer of product designs knows how to evaluate the durability of a product with respect to a Weibull Goal Line taken as defining a life target for the product.

In this bulletin we shall show how to use the concept of Culminated Entropy as a powerful tool in making such evaluations with respect to Weibull Goal Lines.

THE DEFINITION OF ENTROPY

General Definition:

Entropy is a concept which in general is defined to be the Absolute Value of the Natural Logarithm of a Success Probability.
In life tests the success probability is the probability of survival to a specific life target. This probability of survival is known as Reliability. Hence. by definition,

$$\text{ENTROPY} = \left| \ln [\text{RELIABILITY}] \right|$$

THE SPECIAL CASE OF WEIBULL DISTRIBUTIONS

A Two-Parameter Weibull Cumulative Distribution Function is represented mathematically as follows:

$$F(x) = 1 - \text{EXP}[-(x/T)^B]$$

where x = Life (Service Time)

F(x) = Fraction Failed in Time x

B = Weibull Slope

T = Characteristic Life

Webull Reliability:

For an item with a Two-Parameter Weibull Life Distribution ,

$F(x) = 1 - \text{EXP}[-(x/T)^B]$ is the Probability of Failure in time x. So, it follows that the Reliability (Probability of Survival) to time x is 1 - F(x) , i.e.,

$$\text{Reliability to Time } x = R(x) = \text{EXP}[-(x/T)^B]$$

Weibull Entropy:

Since Reliability = $R(x)$ = EXP $[-(x/T)^B]$, it follows that $\ln R(x) = -(x/T)^B$, or

$$\text{ENTROPY to } x = \left| \ln R(x) \right| = (x/T)^B .$$

Thus, an item which has run for time x has accumulated an Entropy equal to $(x/T)^B$.

If B = 1, Entropy grows linearly with service time x.

If B = 2, Entropy grows as the square of service time x.

If B = 3, Entropy grows as the cube of service time x.

Etc. Etc.

THE MEANING OF CULMINATED ENTROPY

Entropy of a Test Sample:

Suppose a test sample on N specimens tells us that these N specimens have running times $x1,x2,x3, \ldots\ldots xN$. Some of these could still be unfailed (suspended) items, while others are actually failed at their indicated life values xi. According to the Weibull Model , the Entropy at xi is $(xi/T)^B$. If xi is a Failure Time, then $(xi/T)^B$ is the Final (Culminated) Entropy of the item under study. On the other hand, if xi is a point of Non-Failure (Suspended Item), then the Culminated Entropy for the item is $(xi/T)^B + 1$. This is because at any unfailed point x, the Average Future Additional Entropy until actual failure is 1 more unit of Entropy.

ADDING UP THE CULMINATED ENTROPIES FOR AN ENTIRE SAMPLE

For a sample of specimens:

Let us define failure times by x1, X2, x3,.....xF
(F = Total No. Failed)

Let us denote unfailed items by U1, U2, U3,.....US
(S = Total No. Suspended)

(F + S = N)

Then, the TOTAL CULMINATED ENTROPY for the entire sample is

$$\mathcal{E}\text{ Sample} = (x1/T)^B + (x2/T)^B + \ldots + (xF/T)^B + (U1/T)^B + (U2/T)^B + \ldots + (US/T)^B + S$$

For a sample of size N, the Culminated Entropy has a Normal Distribution with a Mean of N , and a Sigma (Standard Deviation) of $\sqrt{N}$ (See Figure 1).



DEFINITION OF EVIDENCE

Evidence is defined as the Natural Logarithm of Odds in favor of a claim (of reliability).

Thus, EVIDENCE = ln [C/(1 - C)] ,

where C = Confidence (Probability in favor of claim)
1 - C = Probability Against the Claim

In a Normal Distribution , Evidence is equal to the Z-Score multiplied by $\pi/\sqrt{3}$. This comes out of the Logistic Function approximation for a Normal Area.

Consequently ,

$$\text{ODDS} = \text{EXP (EVIDENCE)} = \text{EXP}\ [(\pi/\sqrt{3}) * Z] ,$$

where , in this case,

$$Z = \frac{\mathcal{E}\,\text{Sample} - N}{\sqrt{N}}$$

For the sample discussed on page 3, this becomes

$$Z = \frac{(x1/T)^B + (x2/T)^B + \ldots + (xF/T)^B + (U1/T)^B + (U2/T)^B + \ldots + (US/T)^B + S - N}{\sqrt{N}}$$

or,

$$Z = \frac{(x1/T)^B + (x2/T)^B + \ldots + (xF/T)^B + (U1/T)^B + (U2/T)^B + \ldots + (US/T)^B - F}{\sqrt{N}}$$

where F = N - S

Once the ODDS are determined, we can calculate

$$\text{CONFIDENCE} = \text{ODDS} / (1 + \text{ODDS})$$

```
10 PRINT TAB(15)"COMPUTER PROGRAM FOR EVALUATING A SAMPLE'S"
20 PRINT TAB(15)"COMPLIANCE TO A TWO-PARAMETER WEIBULL GOAL"
30 PRINT TAB(15)"__________________________________________"
40 PRINT:PRINT:PRINT
50 PRINT TAB(30)"PROGRAM (SAMPCOMP)"
60 PRINT TAB(30)"__________________"
70 PRINT:PRINT
80 DIM X(100)
90 E=0
100 INPUT "GOAL WEIBULL SLOPE = ";B
110 PRINT
120 INPUT "GOAL CHAR. LIFE = ";T
130 PRINT
140 INPUT "TOTAL SAMPLE SIZE = ";N
150 PRINT
160 INPUT "TOTAL NO. FAILED = ";F
170 PRINT
180 FOR I = 1 TO N
185 PRINT "ENTER LIFE VALUE NO. ";I
190 PRINT
195 INPUT X(I)
200 PRINT
210 INPUT "INDEX (0 FOR SUSP., 1 FOR FAILED) = ";K
220 PRINT
230 IF K = 0 THEN PRINT "SUSP. ITEM AT ";X(I)
240 IF K = 1 THEN PRINT "FAILURE AT ";X(I)
250 E = E + ((X(I))/T)^B
255 PRINT
260 NEXT I
265 CLS
270 PRINT
280 PRINT TAB(20)"TOTAL SAMPLE SIZE = ";N
290 PRINT
300 PRINT TAB(20)"TOTAL NO. FAILED = ";F
310 PRINT
320 Z = (E - F)/SQR(N)
330 W = EXP((3.1415927# * Z)/SQR(3))
340 C = W/(1 + W)
350 PRINT TAB(20)"GOAL WEIBULL SLOPE = ";B
360 PRINT
370 PRINT TAB(20)"GOAL CHAR. LIFE = ";T
380 PRINT
390 PRINT TAB(20)"CONFIDENCE OF COMPLIANCE = ";C
400 PRINT
500 END
```

ACTUAL EXAMPLE USING THE COMPUTER PROGRAM

Suppose we test 12 items, and that there are 7 failures at 502 hrs., 300 hrs., 850 hrs., 1245 hrs., 610 hrs., 700 hrs., and 935 hrs. Furthermore, suppose there are also 5 non-failures at 150 hrs., 400 hrs., 1000 hrs., 350 hrs., and 525 hrs.

Let the GOAL LINE be F(x) = 1 - EXP[-(x/720)^2.5]

Then the SAMPCOMP printout turns out to be as follows :

COMPUTER PROGRAM FOR EVALUATING A SAMPLE'S COMPLIANCE TO A TWO-PARAMETER WEIBULL GOAL

PROGRAM (SAMPCOMP)

FAILURE AT 502
FAILURE AT 300
FAILURE AT 850
FAILURE AT 1245
FAILURE AT 610
FAILURE AT 700
FAILURE AT 935
SUSP. ITEM AT 150
SUSP. ITEM AT 400
SUSP. ITEM AT 1000
SUSP. ITEM AT 350
SUSP. ITEM AT 525

TOTAL SAMPLE SIZE = 12
TOTAL NO. FAILED = 7
GOAL WEIBULL SLOPE = 2.5
GOAL CHAR. LIFE = 720
CONFIDENCE OF COMPLIANCE = .9499266

CONCLUSION

The "SAMCOMP" program is very handy and effective in evaluating how a sample of test times (some failed and some suspended) complies with a Two-Parameter Weibull Goal Line. The program itself is so constructed that it is not necessary to put the test lives in any particular order before starting with the input data. The program simply asks the analyst to enter one item at a time, together with an index 0 or 1, where 0 is used for suspended items, and index 1 indicates a failed item. Then the program totals up all of the Culminated Entropies from the data points and calculates the Confidence that the sample of test lives complies with the Weibull Goal Line as a whole. In this way, the analyst quickly can tell how well a design is doing with respect to its durability goal.